# Maintainer: Massimiliano Torromeo <massimiliano.torromeo@gmail.com>
# Maintainer: Maxim Baz <$pkgname at maximbaz dot com>

pkgname=systembus-notify
pkgver=1.1
pkgrel=2
pkgdesc="System bus notification daemon"
arch=('loong64' 'x86_64')
url="https://github.com/rfjakob/systembus-notify"
license=('MIT')
depends=('systemd')
backup=("etc/xdg/autostart/systembus-notify.desktop")
source=("https://github.com/rfjakob/$pkgname/archive/v$pkgver/$pkgname-$pkgver.tar.gz")
sha256sums=('267f507811940f5e4129847e38ce54f6e40f66af5cbb6f46e84009d5f9a5afd6')

build() {
    cd "$pkgname-$pkgver"
    make
}

package() {
    cd "$pkgname-$pkgver"
    install -Dm755 -t "$pkgdir/usr/bin/" systembus-notify
    install -Dm644 -t "$pkgdir/etc/xdg/autostart/" systembus-notify.desktop
    install -Dm644 -t "$pkgdir/usr/share/licenses/$pkgname/" LICENSE
}
